
Problem-2.8. Dizinin Elemanlarını İşaretine Göre Ayırma 

Negatif, pozitif ve sıfır değerli elemanlardan oluşan bir x dizisinin pozitif elemanlarını y dizisine, 

negatif elemanlarını z dizisine ve sıfır değerli elemanlarını t dizisine aktaran programı yazalım. Verilen 

n boyutlu bir x dizisinin elemanları i=1, 2, …, n olmak üzere, 

x(i)<0 ise negatif,  

x(i)>0 ise pozitif, 

x(i)=0 ise sıfır, 

değerlidir denir. 

 

INPUT "dizinin eleman sayısı=", n 

DIM x(n), y(n), z(n), t(n) 

FOR i = 1 TO n 

    PRINT "x("; i; ")=", 

    INPUT x(i) 

NEXT i 

FOR i = 1 TO n 

    PRINT x(i); 

NEXT i 

k = 0: l = 0: m = 0 

FOR i = 1 TO n 

    IF x(i) > 0 THEN k = k + 1: y(k) = x(i) 

    IF x(i) < 0 THEN l = l + 1: z(l) = x(i) 

    IF x(i) = 0 THEN m = m + 1: t(m) = x(i) 

NEXT i 

PRINT : PRINT 

PRINT "y-dizisi" 

FOR i = 1 TO k 

    PRINT y(i); 

NEXT i 

PRINT : PRINT 

PRINT "z-dizisi" 

FOR i = 1 TO l 

    PRINT z(i); 

NEXT i 

PRINT : PRINT 

PRINT "t-dizisi" 

FOR i = 1 TO m 

    PRINT t(i); 



NEXT i 

Problem-2.9. Dizinin Artan (Azalan) Olup Olmadığının İncelenmesi 

Verilen bir dizinin artan (azalan) olup olmadığını bulan programı yazalım. i  için verilen bir 

xi dizisinde, her i=1,…,n-1 için xi<xi+1 (xi>xi+1) koşulu sağlanıyor ise bu diziye artan (azalan) dizi denir. 

Buna göre bir dizinin artan veya azalan dizi olup olmadığını bulacak programlar aşağıdaki gibi 

yazılabilir. 

 

INPUT "n=", n 

DIM x(n) 

FOR i = 1 TO n 

    PRINT "x("; i; ")=", 

    INPUT x(i) 

NEXT i 

sayac = 0 

FOR i = 1 TO n - 1 

    IF x(i) < x(i + 1) THEN  

        sayac = sayac + 1 

    ELSE 

        EXIT FOR 

    END IF 

NEXT i 

IF sayac = n - 1 THEN 

    PRINT "dizi artandır" 

ELSE 

    PRINT "dizi artan değildir" 

END IF 

'Verilen bir dizinin azalan olup olmadığını bulan program 

INPUT "n=", n 

DIM x(n) 

FOR i = 1 TO n 

    PRINT "x("; i; ")=", 

    INPUT x(i) 

NEXT i 

sayac = 0 

FOR i = 1 TO n - 1 

    IF x(i) > x(i + 1) THEN  

        sayac = sayac + 1 

    ELSE 



        EXIT FOR 

    END IF 

NEXT i 

IF sayac = n - 1 THEN 

    PRINT "dizi azalandır" 

ELSE 

    PRINT "dizi azalan değildir" 

END IF 

Problem-2.10. Dizinin Her Bir Elemanından Kaçar Tane Olduğunun Bulunması 

1’den n’e kadar tamsayılar içeren m boyutlu bir dizide her elemandan kaç tane olduğunu bulan 

programı yazalım. Örneğin 1’den 10’a kadar tamsayıları içeren 20 boyutlu x dizisini göz önüne alalım. 

x dizisinin elemanları, 1, 2, 3, 7, 6, 5, 9, 10, 8, 4, 1, 6, 4, 3, 9, 2, 8, 2, 6, 9 olarak verilsin. x dizisinin her 

bir elemanından kaç tane olduğu Tablo 2.1.’de verilmektedir. 

 

Tablo 2.1. x dizisinin her bir elemanından kaçar tane olduğu. 

INPUT "n=", n 

INPUT "Dizinin eleman sayısı=", m 

DIM x(m) 

FOR i = 1 TO m 

    INPUT x(i) 

NEXT i 

FOR i = 1 TO n 

    s = 0 

    FOR j = 1 TO m 

        IF x(j) = i THEN s = s + 1 

    NEXT j 

    IF s <> 0 THEN PRINT i; "elemanından "; s; "tane var" 

NEXT i 



n=10 

Dizinin eleman sayısı=20 

x( 1 )=? 1 :x( 2 )=? 2 : x( 3 )=? 3 : x( 4 )=? 7 : x( 5 )=? 6 : x( 6 )=? 5 

x( 7 )=? 9: x( 8 )=? 10: x( 9 )=? 8: x( 10 )=? 4: x( 11 )=? 1: x( 12 )=? 6 

x( 13 )=? 4: x( 14 )=? 3: x( 15 )=? 9: x( 16 )=? 2: x( 17 )=? 8: x( 18 )=?     

2: x( 19 )=? 6 : x( 20 )=? 9 

 1 elemanından  2 tane var 

 2 elemanından  3 tane var 

 3 elemanından  2 tane var 

 4 elemanından  2 tane var 

 5 elemanından  1 tane var 

 6 elemanından  3 tane var 

 7 elemanından  1 tane var 

 8 elemanından  2 tane var 

 9 elemanından  3 tane var 

10 elemanından  1 tane var 

 


